Creating Parameter Objects: The Basics

Introduction

Parameter objects are simply a construct that allows you to pass multiple parameters into a method call using one object to do it. This approach simplifies the method’s signature by reducing the number of parameters passed into the method from an itemized list to only one. Why should you want to do this? The following is an example of the same method call. The first example uses the old method of itemizing the parameters. The second example utilizes a parameter object to accomplish the same thing.

Listing 1: Code Snippet using Itemized Parameter List

goApp.SomeMethod( <cUserName> , ;
<cUserPassword> , ;
<dLoginDate> , ;
<cLoginTime> , ;
<lAdmin )

Listing 2: Example of method call using Itemized Parameter List

goApp.SomeMethod( "CTB","CTB",DATE(),TIME(),.T.)

Listing 3: Example of the same method call using a Parameter Object

loParameter = CREATEOBJ("prmParameter")
WITH loParameter
.cUserName = "CTB"
.cUserPassword = "CTB"
.dLoginDate = DATE()
.cLoginTime = TIME()
.lAdmin = .T.
ENDWITH

goApp.SomeMethod( loParameter )

It becomes readily apparent that using a parameter object provides you with the following benefits. First, it simplifies the method call by reducing it to a single parameter. This, in itself, means two things: 1) developers no longer have to remember the order in which to pass parameters into a method (easier to use) and 2) it allows for additional parameters to be passed into the method without changing the method’s signature (easier to maintain). Second, the code is much more self-documenting. For example, in Listing 2, is it possible for a maintenance programmer to know that the first "CTB" is a user name and the second is the password? However, in Listing 3 it is quite apparent what each parameter does. Notice also that the order in which the parameter object’s properties are provided is no longer important. The following code works just as well as that provided in Listing 3.

loParameter = CREATEOBJ("prmParameter")
WITH loParameter
.cUserPassword = "CTB"
.dLoginDate = DATE()
.lAdmin = .T.
.cUserName = "CTB"
.cLoginTime = TIME()
ENDWITH

Creating the example parameter object

Step One: Creating the Parameter Object’s Class Definition

All of COMCodebook’s parameter objects are derived from the superclasses contained in the IPARAMETEROBJECTS.PRG. This program is located in the \CODEBOOK\COMMON\ILIBS\ subdirectory and it contains the IAbstractParameter class. This is the class you should use to create all of your application’s parameter objects. You should place all of your application’s parameter objects in an A Layer program. Therefore, create a program named APARAMETEROBJECTS.PRG and place it in your application’s LIBS subdirectory. Next, create the following class definition therein:

DEFINE CLASS prm_TestParameter AS IAbstractParameter

ENDDEFINE

Step Two: Add the Parameters to the Parameter Object

Parameters become part of the parameter object by adding custom properties to the class definition. In our case, we need to store information about five parameters so we create five custom properties to store the values. To do this, modify the class definition as follows:

DEFINE CLASS prm_TestParameter AS IAbstractParameter

cUserPassword = ""
cUserName = ""
dLoginDate = DATE()
cLoginTime = TIME()
lAdmin = .F.

ENDDEFINE

Step Three: Teach the parameter object about its parameters

Visual Foxpro contains no functions that allow an object to know about the nature, type or count of its custom properties. In other words, there is no innate VFP method to ask an object the following: 

1. How many custom properties do you have? 

2. What are their names? 

3. What are their types?

The parameter object uses a member array along with some custom methods to store this information and answer these questions. 

The last step required to create a parameter object is to override the DefineParameterArray() method by using the AddParameter() method to teach the parameter object about the five parameters for which it is responsible. Modifying the above class definition as follows completes the creation of the parameter object required to power the fictitious goApp.SomeMethod() method.

DEFINE CLASS prm_TestParameter AS IAbstractParameter

cUserPassword = ""
cUserName = ""
dLoginDate = DATE()
cLoginTime = TIME()
lAdmin = .F.

PROCEDURE DefineParameterArray
DODEFAULT()
THIS.AddParameter( "cUserName" , "C" )
THIS.AddParameter( "cUserPassword" , "C" )
THIS.AddParameter( "dLoginDate" , "D" )
THIS.AddParameter( "cLoginTime" , "C" )
THIS.AddParameter( "lAdmin" , "L" )
ENDPROC

ENDDEFINE

Testing the new Parameter Object

To test your new parameter object you must first set the libraries needed to support the object. They are as follows:

SET PROC TO ..\..\libs\csession.prg
SET PROC TO ..\..\ilibs\isession.prg ADDITIVE
SET PROC TO ..\..\libs\cparameterobjects.prg ADDITIVE
SET PROC TO ..\..\ilibs\iparameterobjects.prg ADDITIVE
SET PROC TO ..\libs\aparameterobjects.prg ADDITIVE

Next, create the object as you normally would any other object.

loParameter = CREATEOBJ('prm_TestParameter')

Finally, populate the parameter object with non-default values.

loParameter.cUserName = "CTB"
loParameter.cPassword = "CTB"
loParameter.lAdmin = .T.

A quick look at the Locals window displays the nature of the newly created parameter object.


Notice that the parameter object contains an array that defines the name and type of each parameter for which it is responsible. We provided this information by overriding the DefineParameterArray() method.

It is very important that you realize the power of this approach. Now that you have a parameter object you can do just about anything with it that you want … it is fully programmable. For example, let’s say you didn’t want the programmer to have to hard code the User ID, Password and Administrative rights each time. You could create methods on your application named GetUserID(), GetUserPassword() and GetAdminRights(). Then you can add these method calls to the parameter object’s class definition as follows:

DEFINE CLASS prm_TestParameter AS IAbstractParameter

cUserName = goApp.GetUserName()
cUserPassword = goApp.GetUserPassword()
dLoginDate = DATE()
cLoginTime = TIME()
lAdmin = goApp.GetUserAdminRights()

PROCEDURE DefineParameterArray
  DODEFAULT()
  WITH This
     .AddParameter( "cUserName" , "C" )
     .AddParameter( "cUserPassword" , "C" )
     .AddParameter( "dLoginDate" , "D" )
     .AddParameter( "cLoginTime" , "C" )
     .AddParameter( "lAdmin" , "L" )
ENDPROC

ENDDEFINE

Now each time the parameter object is created there is nothing for the programmer to do but use it in the method call. Basically, you can implement your design anyway you see fit. 

Conclusion

Using parameter objects provides several benefits, method signatures that are maintainable (read flexible) over time, self-documenting code, and the capability to programmatically enhance the parameter itself. COMCodebook makes use of several complex parameter objects. The first, CAbstractADOConfigurationParameter, models the processes required to create connection strings for ADO. The second, CADOComponentsParameter, models ADO itself by encapsulating the whole ADO Object Model within the parameter object. I’ll discuss these more advanced parameter objects in another white paper soon to follow.

 

